



Function Calls

Branch with Link (function call) instruction

bl label



- Function call to the instruction with the address label (no local labels for functions)
 - imm24 number of instructions from pc+8 (24-bits)
 - label any function label in the current file, any function label that is defined as .global in any file that it is linked to, any C function that is not static

Branch with Link Indirect (function call) instruction

blx Rm



- Function call to the instruction whose address is stored in Rm (Rm is a function pointer)
- bl and blx both save the address of the instruction immediately following the bl or blx instruction in register
 Ir (link register is also known as r14)
- The contents of the link register is the <u>return address in the calling function</u>
 - (1) Branch to the instruction with the label f1
- (2) copies the address of the instruction AFTER the bl in Ir



Function Call Return

Branch & exchange (function return) instruction


bx 1r



// we will always use lr


- Causes a branch to the instruction whose address is stored in register <1r>
 - It copies 1r to the PC
- This is often used to implement a return from a function call (exactly like a C return) when the function is called using either bl label, or blx Rm

push: Multiple Register Save to the stack



- push copies the contents of the {reg list} to stack segment memory
- push subtracts (# of registers saved) * (4 bytes) from the sp to allocate space on the stack
 - sp = sp (# registers saved * 4)
- this must always be true: sp % 8 == 0

pop: Multiple Register Restore from the stack



- pop copies the contents of stack segment memory to the {reg list}
- pop <u>adds:</u> (# of registers restored) * (4 bytes) to <u>sp</u> to <u>deallocate</u> space on the stack
 - sp = sp + (# registers restored * 4)
- Remember: {reg list} must be the same in both the push and the corresponding pop

Registers: Rules For Use

Register	Function Call Use	Function Body Use	Save before use Restore before return
r0	arg1 and return value	scratch registers	No
r1-r3	arg2 to arg4	scratch registers	No
r4-r10	preserved registers	contents preserved across function calls	Yes
r11 / fp	stack frame pointer	Use to locate variables on the stack	Yes
r12 / ip	may used by assembler with large text file	can be used as a scratch if really needed	No
r13 / sp	stack pointer	pointer stack space allocation	
r14 / lr	link register	nk register contains return address for function calls	
r15	Do not use	Do not use	No

Return Value and Passing Parameters to Functions

(Four parameters or less)

Register	Function Call Use	Function Body Use	Save before use Restore before return
r0	arg1 and return value	scratch registers	No
r1-r3	arg2 to arg4	scratch registers	No

• Where r0, r1, r2, r3 are arm registers, the function declaration is (first four arguments):

- Each parameter and return value is limited to data that can fit in 4 bytes or less
- Calling function:
 - · copy up to the first four parameters into these four registers before calling a function
 - MUST assume that the called function will alter the contents of all four registers: r0-r3
 - In terms of C runtime support, these registers contain the copies given to the called function
 - C allows the copies to be changed in any way by the called function
- For parameters, whose size is larger than 4 bytes, pass a pointer to the parameter (we will cover this later)
- · Called function:
 - you receive the first four parameters in these four registers (r0 r3)

What it means to be a Temporary/argument register

```
int a(void)
{
    // not shown
}
int main(void)
{
    int r0 = 0;
    int r1 = 1;
    int r2 = 2;
    int r3 = 3;
    r0 = a();
    // in C r1 and r3 would have the same values
    // after the call
```

```
// main()
// code not shown
mov r0, 0
mov r1, 1
mov r2, 2
mov r3, 3
bl a
// r0 = return value
// r1-r3 values are unknown as a() has right to change them as it wants
```

Preserved Registers

Register	Function Call Use	Function Body Use	Save before use Restore before return
r4-r10	preserved registers	contents preserved across function calls	Yes
r11/fp	stack frame pointer Use to locate variables on the stack		Yes
r13/sp	stack pointer	stack space allocation	Yes
r14/lr	link register	contains return address for function calls	Yes

- Any value you have in a preserved register before a function call will still be there after the function returns (Contents are "preserved" across function calls)
- If the function wants to use a preserved register it must:
 - 1. Save the value contained in the register at function entry
 - 2. Use the register in the body of the function
 - 3. Restore the original saved value to the register at function exit (before returning to the caller)
- You use a preserved register when a function makes calls another function and you have:
 - 1. Local variables allocated to be in registers
 - 2. Parameters passed to you (in r0-r3) that you need to continue to use after calling another function

Minimum Stack Frame (Arm Arch32 Procedure Call Standards)

• Minimal frame: allocating at function entry: push {fp, lr}


Minimum stack frame



- sp always points at top element in the stack (lowest byte address)
- fp always points at the bottom element in the stack
 - Bottom element is always the saved 1r (contains the return address of caller)
 - A saved copy of callers fp is always the next element below the Ir
 - fp will be used later when referencing stack variables
- Minimal frame: deallocating at function exit: pop {fp, lr}
- On function entry: sp must be 8-byte aligned (sp % 8 == 0)

First Look: A typical Stack Frame

- Saved Ir and fp of the caller (so function calls work)
- Save values for any preserved registers this function will change
- Space (FRMADD) for local variables is allocated on the stack right below the lowest pushed register



X

12

Function Prologue and Epilogue


```
myfunc:
  Function
                        {fp, lr}
                push
                                               // push (save) fp and lr on stack
  Prologue
                add
                        fp, sp, FP_OFF
                                               // set fp for this function
   creates
                                               // allocate FRMADD bytes for local vars
                add
                        sp, sp, -FRMADD
stack frame
                                               // by moving sp
                   // your code
  Function
                        sp, fp, FP_OFF
                                               // deallocate local variables by moving sp
                sub
  Epilogue
                                               // pop (restore) fp and lr from stack
                        {fp, lr}
                pop
  removes
                                               // return to caller
                bx
                        lr
stack frame
```

 x

Minimum Stack Frame (Arm Arch32 Procedure Call Standards)

main() calls funcA() saved Ir Function entry (Function Prologue): main() allocate stack space callers fp save Ir and fp registers (push) SP = SP - "space"Ir to main(grows "down" fp set fp to top entry in stack main() fp allocate space for local vars – later slides • Function return (Function **Epilogue**): deallocate stack space 4 deallocate space for locals -later fp saved Ir SP = SP + "space" main(restores Ir and fp registers (pop) shrinks "up" callers fp sp Return To Caller 3.

Function Prologue: Allocating the Stack Frame -1




Function Prologue: Allocating the Stack Frame - 2 Prologue Step 3 of 3

- Space for local variables is allocated on the stack right below the lowest pushed register
- Add memory to the stack frame for local variables by moving the sp towards low memory
- The amount moved is the total size of all local variables in bytes plus memory alignment padding

FRMADD = total local var space (bytes) + padding

Allocate the space after the register push by


 fp (frame pointer) is used as a pointer (base register) to access all stack variables — later slides




Allocate Space for Local Variables

```
push {fp, lr} // push (save) fp and lr on stack
add fp, sp, FP_OFF // set fp for this function
add sp, sp, -FRMADD // allocate FRMADD bytes for local vars
// by moving sp
```

Function Epilogue: Deallocating the Stack Frame - 1




Use fp as a pointer to find local variables on the stack





Move SP back to where it was after the push in the prologue. So, pop works properly (this also deallocates the local variables)

```
sub sp, fp, FP_OFF // deallocate local variables by moving sp pop {fp, lr} // pop (restore) fp and lr from stack bx lr // return to caller
```

Why You must move SP before POP in the Epilogue








```
sub sp, fp, FP_OFF // deallocate local variables by moving sp pop {fp, lr} // pop (restore) fp and lr from stack // return to caller
```

Function Epilogue: Deallocating the Stack Frame

Step 1 Deallocate Space for locals Put SP back so pop works sub sp, fp, FP OFF saved Ir callers fp saved Ir callers fp Stack FP OFF Frame saved r5 saved r4 sp Local Deallocated Variables stack space low memory

Move SP back to where it was after the push in the prologue. So, pop works properly (this also deallocates the local variables)


19



Use pop to restore the registers to the values they had at function entry

```
sub sp, fp, FP_OFF // deallocate local variables by moving sp pop {fp, lr} // pop (restore) fp and lr from stack bx lr // return to caller
```

How to Set FP



# regs saved	FP_OFF in Bytes Distance from Ir to lowest saved register
2	4
3	8
4	12
5	16
6	20
/// /	24
8	28
9	32

 $FP_OFF = (\#regs saved - 1) * 4$

grows down

Means Caution, odd number of saved regs! If odd number pushed, make sure frame is 8byte aligned (later)

this must always be true: sp % 8 == 0

fp = sp + 20

bvtes

4-byte words

Reference Table: Global Variable access

var		global variable contents	contents of r0 into	
7 (4)	into r0 (Iside)	into r0 (rside)	global variable	
x	ldr r0, =x	ldr r0, =x ldr r0, [r0]	ldr r1, =x str r0, [r1]	
*x	ldr r0, =x ldr r0, [r0]	ldr r0, =x ldr r0, [r0] ldr r0, [r0]	ldr r1, =x ldr r1, [r1] str r0, [r1]	
**X	ldr r0, =x ldr r0, [r0] ldr r0, [r0]	ldr r0, =x ldr r0, [r0] ldr r0, [r0] ldr r0, [r0]	ldr r1, =x ldr r1, [r1] ldr r1, [r1] str r0, [r1]	
stderr	ldr r0, =stderr	ldr r0, =stderr ldr r0, [r0]	<pre><do are="" doing="" know="" not="" really="" unless="" what="" write="" you=""></do></pre>	
.Lstr	ldr r0, =.Lstr	ldr r0, =.Lstr ldrb r0, [r0]	<read only=""></read>	

```
.bss // from libc
stderr:.space 4 // FILE *
```

```
.data
x: .data y //x = &y
```

```
.section .rodata
.Lstr: .string "HI\n"
```

stdin, stdout and stderr are global variables

Assembler Directives: Label Scope Control (Normal Labels only)

```
.extern printf
.extern fgets
.extern strcpy
.global fbuf
```

.extern <label>

- Imports label (function name, symbol or a static variable name);
- An address associated with the label from another file can be used by code in this file

```
.global <label>
```

- Exports label (or symbol) to be visible outside the source file boundary (other assembly or c source)
- label is either a function name or a global variable name
- Only use with function names or static variables
- Without .global, labels are usually (depends on the assembler) local to the file

Passing global variables as a parameter: fprintf()

```
    r0 = function(r0, r1, r2, r3)
        fprintf(stderr, "arg2", arg3, arg4)
    create a literal string for arg2 which tells fprintf() how to interpret the remaining arguments
    stdin, stdout, stderr are all global variable and are part of libc
    these names are their lside (label names)
    get their contents and pass that to fprintf(), fread(), fwrite()
```

```
.extern fprintf //declare fprintf
#include <stdio.h>
                                                  .section .rodata // note the dots "."
#include <stdlib.h>
                                           .Lfst: .string "c=%d\n"
int
                  We are going to
main(void)
                  put these
{
                  variables in
                                          // part of the text segment below
   int a = 2;
                                                         r2, 2 // int a = 2;
                                                  mov
                  temporary
   int b = 3;
                                                         r3, 3 // int b = 3;
                                                  mov
                  registers
   int c;
                                                         r2, r2, r3 // arg 3: int c = a + b;
                                                  add
    c = a + b;
                                                         r0, =stderr // get stderr address
                                                  ldr
   fprintf(stderr, "c=%d\n", c);
                                  three passed
                                                  ldr
                                                         r0, [r0] // arg 1: get stderr contents
                                  args in this
                                                         r1, =.Lfst // arg 2: =literal address
                                                  ldr
            r0, r1,
                          r2
                                  use of fprintf
                                                  bl
                                                          fprintf
    return EXIT SUCCESS;
```

Example: using preserved registers for local variables

```
#include <stdio.h>
#include <stdlib.h>
int
both getchar() and
putchar() alter r0-r3

int c; // use r0
int count = 0; // use r4


r0

while ((c = getchar()) != EOF) {

   putchar(c);
   count++;
   }

printf("Echo count: %d\n", count);
   return EXIT_SUCCESS;
}
```

```
.extern getchar
        .extern putchar
        .section .rodata
       .string "Echo count: %d\n"
.Lst:
        .text
               main, %function
        .type
        .global main
               EOF,
        .equ
                             -1
        .equ FP OFF,
                             12
               EXIT SUCCESS, 0
        .equ
main:
              {r4, r5, fp, lr}
        push
              fp, sp, FP OFF
       add
               r4, 0 //r4 = count
       mov
/* while loop code will go here */
              r0, EXIT SUCCESS
       mov
               sp, fp, FP OFF
       sub
               {r4, r5, fp, lr}
       pop
       bx
               1r
        .size main, (. - main)
```



Accessing Pointers (argv) in ARM assembly

```
% ./cipher -e -b in/B00K
   .extern printf
                                                                               argv[0] = ./cipher
   .extern stderr
   .section .rodata
                                                                               argv[1] = -e
.Lstr: .string "argv[%d] = %s\n"
                                                                               argv[2] = -b
   .text
                                               need to save r1 as
   .global main // main(r0=argc, r1=argv)
                                                                               argv[3] = in/B00K
           main, %function
   .tvpe
                                               we are calling a
           FP_OFF,
   •equ
                                               function - fprintf
main:
                                                                                   r0-r3 lost due to fprintf call
           {r4-r7, fp, lr}
   push
           fp, sp, FP_OFF
   add
                           // save argv!
           r7, r1
   mov
           r4, =stderr
                          // get the address of stderr
   ldr
                                                                                           "argv[%d] = %s\n"
           r4, [r4]
                          // get the contents of stderr
   ldr
                                                                   Registers
                         // get the address of .Lstr
   ldr
           r5, =.Lstr
   mov
           r6. 0
                           // set indx = 0:
                                                             r7
                                                                                            NULL
.Lloop:
                                                             r6
                                                                      indx
                                                                                          argv[3]
   // fprintf(stderr, "argv[%d] = %s\n", indx, *argv)
   ldr
           r3, [r7]
                          // arg 4: *argv
                                                                                          argv[2]
           r3, 0
                          // check *argv == NULL
   cmp
                                                                                          argv[1]
           Ldone
   beg
                          // if so done
                                                                  file * stderr
           r2, r6
                          // arg 3: indx
   mov
                                                                                          argv[0]
                          // arg 2: "argv[%d] = %s\n"
                                                             r3
           r1, r5
   mov
           r0. r4
                          // arg 1: stderr
   mov
   bl
           fprintf
                                                             r2
           r6, r6, 1
                          // indx++ for printing
   add
           r7, r7, 4
                          // argv++ pointer
   add
                                                                     **argv
   b
           .Lloop
.Ldone:
                              observe the
                                                                      argc
           r0, 0
   mov
                              different
           sp, fp, FP_OFF
   sub
           {r4-r7, fp, lr}
   pop
                              increment sizes
                                                            fprintf(stderr, "argv[%d] = %s\n", indx, *argv);
   bx
```

in/book

-b

-e

./cipher

Allocating Space For Locals on the Stack


- Space for local variables is allocated on the stack right below the lowest pushed register
 - Move the sp towards low memory by the total size of all local variables in bytes plus padding

FRMADD = total local var space (bytes) + padding

Allocate the space after the register push by

Requirement: on function entry, sp is always 8-byte aligned
 sp % 8 == 0

- · Padding (as required):
 - 1. Additional space between variables on the stack to meet memory alignment requirements
 - 2. Additional space so the frame size is evenly divisible by 8
- fp (frame pointer) is used as a pointer (base register) to access all stack variables — later slides



27

Local Variables on the stack

```
after push {r4-r5,fp,lr}
                                add fp, sp, FP OFF
int main(void)
                                     saved Ir
                                     callers fp
    int c;
                                     saved r5
    int count = 0;
    // rest of code
                                     saved r4
                                                      sp after push
                          FRMADD
                          = 8
                                      count
                                                      sp after
                                                      allocating locals
```

```
.text
       .tvpe
               main, %function
       .global main
              FP OFF,
      .equ
                         12
              FRMADD,
                          8
      .equ
   main:
              {r4, r5, fp, lr}
       push
               fp, sp, FP OFF
       add
               sp, sp, -FRMADD
       add
   // but we are not done yet!
// when FRMADD values fail to assemble
        ldr r3, =-FRMADD
```

- In this example we are allocating two variables on the stack
- When writing assembly functions, in many situations you may choose allocate these to registers instead

- Add space on the stack for each local
 - we will allocate space in same order the locals are listed the C function shown from high to low stack address

add sp, sp, r3

- gcc compiler allocates from low to high stack addresses
- Order does not matter for our use

28

Accessing Stack Variables: Introduction

```
int main(void)
{
    int c;
    int count = 0;
    // rest of code
}
```

```
4 saved lr

24 bytes

8-bytes

8-byte aligned

12 saved r5

12 saved r4

4 count

Sp

14 saved lr

4 callers fp

4 count

15 saved r4

4 count

16 saved r4

17 saved r4

18 saved r4

19 saved r4

10 saved r4

10 saved r4

11 saved r5

12 saved r4

13 saved r4

14 count

15 saved r4
```

- To Access data stored in the stack
 - use the ldr/str instructions
- Use register fp with offset (distance in bytes)
 addressing (use either register offset or immediate offset)
- No matter what address the stack frame is at, fp always points at saved lr, so you can find a local stack variable by using an offset address from the contents of fp


Variable	distance from fp	Read variable	Write Variable
int c	-16	ldr r0, [fp, -16]	str r0, [fp, -16]
int count	-20	ldr r0, [fp, -20]	str r0, [fp, -20]

low memory 4-byte words


```
.text
    .type
            main, %function
    .global main
           FP OFF,
                      12
   .equ
           FRMADD,
                       8
   .equ
main:
            {r4, r5, fp, lr}
    push
            fp, sp, FP_OFF
    add
            sp, sp, -FRMADD
    add
// but we are not done vet!
```

Stack Frame Design – Local Variables

- When writing an ARM equivalent for a C program, for CSE30 we will not re-arrange the order of the variables to optimize space (covered in the compiler course)
- Arrays start at a 4-byte boundary (even arrays with only 1 element)
 - Exception: double arrays [] start at an 8-byte boundary
 - struct arrays are aligned to the requirements of largest member
- Single chars (and shorts) can be grouped together in same 4-byte word (following the alignment for the short)
- Padding may be required (see next slide)




Rule: When the function is entered the stack is already 8-byte aligned



30^L


Stack Variables: Padding

 Variable padding – start arrays at 4byte boundary and leave unused space at end (high side address) before the variable higher on the stack



 Frame padding – add space below the last local variable to keep 8-byte alignment






```
int main(void)
{
    int c;
    int count = 0;
    char buf[] = "hi";
    // rest of code
}
```

```
.text
    .type
            main, %function
    .global main
   .equ
           FP OFF,
                      12
           FRMADD,
                      16
   .equ
main:
            {r4, r5, fp, lr}
    push
            fp, sp, FP_OFF
    add
    add
            sp, sp, -FRMADD
// but we are not done yet!
```

Accessing Stack Variables, the hard way.


```
int main(void)
{
    int c;
    int count = 0;
    char buf[] = "hi";
    // rest of code
}
```

```
.text
            main, %function
    .type
    .global main
           FP OFF,
   .equ
                       12
           FRMADD,
                       16
   .equ
main:
    push
            {r4, r5, fp, lr}
            fp, sp, FP_OFF
    add
            sp, sp, -FRMADD
    add
// but we are not done yet!
```




char buf[] by usage with ASCII chars we will use strb (or make it unsigned char)


Variable distan		Read variable	Write Variable	
int c	16	ldr r0, [fp, -16]	str r0, [fp, -16]	
int count	20	ldr r0, [fp, -20]	str r0, [fp, -20]	
char buf[0]	24	ldrb r0, [fp, -24]	strb r0, [fp, -24]	
char buf[1]	23	ldrb r0, [fp, -23]	strb r0, [fp, -23]	
char buf[2]	22	ldrb r0, [fp, -22]	strb r0, [fp, -22]	



- Calculating offsets is a lot of work to get it correct
- · It is also hard to debug
- There is a better way!


Best Practice: Assembler Generated FP Distance Table





- 1. For each stack variable create a .equ symbol whose value is the distance in bytes from the FP after the prologue push
- 2. After the last variable add a name PAD for the size of the frame padding (if any) if no padding, PAD will be set to the same value as the variable above it
- 3. The value of the symbol is an expression that calculates the distance from the FP based on the distance of the variable above it on the stack. The first variable will use SP_OFF as the starting distance
 - **.equ VAR**, size_of var + variable_padding + previous_var_symbol // previous_var_symbol distance of the var above
- 4. Calculate the size of the local variable area that needs to be added to the sp in bytes

 FRMADD = distance PAD minus distance of the SP to the FP (FP OFF) after the prologue push



Variable	distance from fp	Address on Stack	Read variable	Write Variable
int c	С	add r0, fp, -C	ldr r0, [fp, -C]	str r0, [fp, -C]
int count	COUNT	add r0, fp, -COUNT	ldr r0, [fp, -COUNT]	str r0, [fp, -COUNT]
char buf[0]	BUF	add r0, fp, -BUF	ldrb r0, [fp, -BUF]	strb r0, [fp, -BUF]
char buf[1]	BUF-1	add r0, fp, -BUF+1	ldrb r0, [fp, -BUF+1]	strb r0, [fp, -BUF+1]
char buf[2]	BUF-2	add r0, fp, -BUF+2	ldrb r0, [fp, -BUF+2]	strb r0, [fp, -BUF+2]

fp -BUF+2

fp –BUF+3


fp -BUF+1

fp -BUF

34

Initializing and Accessing Stack variables

```
.section .rodata
.Lmess: .string "%d %d %s\n"
    .extern printf
    .text
    .type
           main, %function
    .global main
    .equ
            FP_OFF,
                       12
                       4 + FP OFF
    .equ
           С,
                       4 + C
    .equ
           COUNT,
                       4 + COUNT
    .equ
            BUF,
           PAD,
                       4 + BUF
    .equ
                       PAD - FP OFF
    .equ
           FRMADD,
main:
           {r4, r5, fp, lr}
    push
           fp, sp, FP OFF
    add
           sp, sp, -FRMADD
    add
   // nothing to do for C
            r2, 0
    mov
           r2, [fp, -COUNT]
    str
                               passes
           r2, [fp, -BUF+2]
    strb
                               contents of
            r2, 'h'
    mov
                               stack var C
           r2, [fp, -BUF]
    strb
            r2, 'i'
    mov
            r2, [fp, -BUF+1]
    strb
           r0, =.Lmess
                              // arg1
    ldr
           r1, [fp, -C]
    ldr
                              // arg2
           r2, [fp, -COUNT]
    ldr
                             // arg3
           r3, fp, -BUF
                              // arg4
    add
    bl
            printf
```




Variable	distance from fp	Address on Stack	Read variable	Write Variable
int c	С	add r0, fp, -C	ldr r0, [fp, -C]	str r0, [fp, -C]
int count	COUNT	add r0, fp, -COUNT	ldr r0, [fp, -COUNT]	str r0, [fp, -COUNT]
char buf[0]	BUF	add r0, fp, -BUF	ldrb r0, [fp, -BUF]	strb r0, [fp, -BUF]
char buf[1]	BUF-1	add r0, fp, -BUF+1	ldrb r0, [fp, -BUF+1]	strb r0, [fp, -BUF+1]
char buf[2]	BUF-2	add r0, fp, -BUF+2	ldrb r0, [fp, -BUF+2]	strb r0, [fp, -BUF+2]

passes address of a stack variable buf

Stack Frame Design Practice

void func(void)
{
 signed char c;
 signed short s;
 unsigned char b[] = "Stack";
 unsigned char *ptr = &bufl
 // rest of code
}


- 1. Write the variables in C
- 2. Draw a picture of the stack frame
- 3. Write the code to generate the offsets
- 4. create the table to access the variables



Variable	distance from fp	Address on Stack	Read variable	Write Variable
signed char c	С	add r0, fp, -C	ldrsb r0, [fp, -C]	strsb r0, [fp, -C]
signed short s	S	add r0, fp, -S	ldrsh r0, [fp, -S]	strsh r0, [fp, -S]
unsigned char b[0]	BUF	add r0, fp, -B	ldrb r0, [fp, -B]	strb r0, [fp, -B]
unsigned char *ptr	PTR	add r0, fp, -PTR	ldr r0, [fp, -PTR]	str r0, [fp, -PTR]

Working with Pointers on the stack

```
int sum(int j, int k)
    return j + k;
void
testp(int j, int k, int (*func)(int, int),
        int *i)
{
    *i = func(j,k);
    return;
}
int main()
{
    int i;
    int (*pf)(int, int) = sum;
    testp(1, 2, pf, &i);
    printf("%d\n", i);
    return EXIT_SUCCESS;
```



```
.section .rodata
.Lmess: .string "%d\n"
    .extern printf
    .text
    .qlobal main
           main, %function
    .type
    . equ
           FP_OFF, 4
                   4 + FP OFF
    . equ
           PF,
                   4 + I
    ₌equ
                   0 + PF
          PAD,
    •equ
           FRMADD, PAD-FP_OFF
    • equ
// FRMADD = 12 - 4 = 8
```

Variable	distance from fp	Address on Stack	Read variable	Write Variable
int i	I	add r0, fp, -I	ldr r0, [fp, -I]	str r0, [fp, -I]
int (*pf)()	PF	add r0, fp, -PF	ldr r0, [fp, -PF]	str r0, [fp, -PF]

Working with Pointers on the stack

```
.global sum
int
                                                      .type
                                                              sum, %function
sum(int j, int k)
                                                  sum:
                                                      push
                                                              {fp, lr}
    return j + k;
                                                      add
                                                              fp, sp, FP_OFF
                                                               r0, r0, r1
                                                      add
void
testp(int j, int k, int (*func)(int, int),
                                                      sub
                                                              sp, fp, FP_OFF
            int *i)
                                                              {fp, lr}
                                                      pop
                                                      bx
                                                              lr
    *i = func(j,k);
                                                  .size sum, (. - sum)
    return;
}
                                                  .qlobal testp
                                                         testp, %function
                                                  .type
int
                                                          FP OFF, 12
                                                  •equ
main()
                                             testp:
                                                          {r4, r5, fp, lr}
                                                  push
    int i;
    int (*pf)(int, int) = sum;
                                                  add
                                                          fp, sp, FP OFF
                                                          r4, r3
                                                  mov
                                                                           // save i
    testp(1, 2, pf, &i);
                                   r0,r1,r2
                                                          r2
                                                                           // r0=func(r0,r1)
                                                  blx
    printf("%d\n", i);
                                   already set
                                                          r0, [r4]
                                                  str
                                                                           // *i = r0
    return EXIT SUCCESS;
}
                                                          sp, fp, FP OFF
                                                  sub
                                                          {r4, r5, fp, lr}
                                                  pop
                                                  bx
                                                          lr
                                              .size testp, (. - testp)
```

```
Address on Stack
                                              Read variable
                                                                   Write Variable
              distance
  Variable
              from fp
int i
                                           ldr r0, [fp, -I]
                                                                str r0, [fp, -I]
                 Ι
                        add r0, fp, -I
int (*pf)()
                 PF
                        add r0, fp, -PF
                                           ldr r0, [fp, -PF]
                                                                str r0, [fp, -PF]
```


```
.qlobal main
            main, %function
    .type
            FP OFF, 4
    • equ
                    4 + FP OFF
    •equ
            PF,
                    4 + I
    • equ
            PAD.
                    0 + PF
    •eau
            FRMADD, PAD-FP_OFF
    •equ
main:
    push
            {fp, lr}
            fp, sp, FP_OFF
    add
    add
            sp, sp,-FRMADD
            r2. = sum
                             // func address
    ldr
            r1, fp, -PF
                             // PF address
    add
            r2, [r1]
                             // store in pf
    str
            r0, 1
                            // arg 1: 1
    mov
            r1, 2
                             // arg 2: 2
    mov
    ldr
            r2, [fp, -PF]
                            // arg 3: (*pf)()
            r3, fp, -I
                             // arg 4: &I
    add
    hl
            testp
            r0, = Lmess
                            // arg 1: "%d\n"
    ldr
            r1, [fp, -I]
                            // arg 2: i
    ldr
    bl
            printf
    sub
            sp, fp, FP_OFF
            {fp, lr}
    qoq
    hx
            ۱r
.size main, (. - main)
```

Passing More Than Four Arguments – At the point of Call

```
r0 = function(r0, r1, r2, r3, arg5, arg6, ... argn)

arg1, arg2, arg3, arg4, ...
```

- Approach: Increase stack frame size to include space for args# > 4
 - Arg5 and above are in <u>caller's stack frame</u> at the bottom of the stack
- Arg5 is always at the bottom (at sp), arg6 and greater are above it
- One arg value per slot! NO arrays across multiple slots
 - chars, shorts and ints are directly stored
 - Structs (not always), and arrays (always) are passed via a pointer
 - Pointers passed as output parameters contain an address that points at the stack, BSS, data, or heap
- Prior to any function call (and obviously at the start of the called function):
 - 1. sp must point at arg5
 - 2. sp and therefore arg5 must be at an 8-byte boundary,
 - Add padding to force arg5 alignment if needed is placed above the last argument the called function is expecting




Passing More Than Four Arguments – At the point of Call

r0 = function(r0, r1, r2, r3, arg5, arg6, ... argn)

arg1, arg2, arg3, arg4, ...

- Called functions have the right to change stack args just like they can change the register args!
 - Caller must always assume all args including ones on the stack are changed by the caller
- Approach: Extend the stack frame to include enough space for stack arguments for the called function that has the greatest number of args
 - 1. Examine every function call in the body of a function
 - 2. Find the function call with greatest arg count, this determines space needed for outgoing args
 - 3. Add the greatest arg count space as needed to the frame layout
 - 4. Adjust PAD as required to keep the sp 8-byte aligned
- In the calling function prior to making the call you must
 - 1. Evaluate first four args: place the resulting values in r0-r3
 - 2. Evaluate Arg 5 and greater and place the resulting values on the stack




Calling Function Stack Frame: Pass ARG 5 and higher

Rules: At point of call

- 1. OARG5 must be pointed at by sp
- 2. SP must be 8-byte aligned

```
int cnt;
r0 = func(r0, r1, r2, r3, OARG5, OARG6);
```



Variable	distance from fp	Address on Stack	Read variable	Write Variable
int cnt	CNT	add r0, fp, -CNT	ldr r0, [fp, -CNT]	str r0, [fp, -CNT]
int oarg6	OARG6	add r0, fp, -OARG6	ldr r0, [fp, -OARG6]	str r0, [fp, -OARG6]
int oarg5	OARG5	add r0, fp, -OARG5	ldr r0, [fp, -OARG5]	str r0, [fp, -OARG5]

Called Function: Retrieving Args From the Stack


- At function start and before the push{} the sp is at an 8-byte boundary
- Args > 4 in <u>caller's stack frame</u> and arg 5 always starts at fp+4
 - Additional args are higher up the stack, with one "slot" every 4bytes

.equ ARGN,
$$(N-4)*4$$
 // where n must be > 4

- This "algorithm" for finding args was designed to enable variable arg count functions like printf("conversion list", arg0, ... argn);
- No limit to the number of args (except running out of stack space)

Rule:

Called functions always access stack args using a positive offset to the fp



Called Function: Retrieving Args From the Stack


```
FP OFF,
.equ
       C, 4 + FP OFF
.equ
    INDX, 4 + C
.equ
       PAD,
                4 + INDX
.equ
       FRMADD, PAD - FP OFF
.equ
// below are distances to caller's stack frame
      ARG6,
                 8
.equ
       ARG5,
.equ
```

$$r0 = func(r0, r1, r2, r3, r4, ARG5, ARG6);$$

Rule:


Called functions always access stack args using a positive offset to the fp

using	a positiv	re offset to the fp		
Variable or Argument	distance from fp	Address on Stack	Read variable	Write Variable
int arg6	ARG6	add r0, fp, ARG6	ldr r0, [fp, ARG6]	str r0, [fp, ARG6]
int arg5	ARG5	add r0, fp, ARG5	ldr r0, [fp, ARG5]	str r0, [fp, ARG5] 💠
int c	С	add r0, fp, -C	ldr r0, [fp, -C]	str r0, [fp, -C]
int count	INDX	add r0, fp, -INDX	ldr r0, [fp, -INDX]	str r0, [fp, -INDX]



Previous


Example: Passing Stack Args, Calling Function



Variable or Argument	distance from fp	Address on Stack	Read variable	Write Variable
int i	I	add r0, fp, -I	ldr r0, [fp, -I]	str r0, [fp, -I]
int (*pf)()	PF	add r0, fp, -PF	ldr r0, [fp, -PF]	str r0, [fp, -PF]
int oarg6	OARG6	add r0, fp, -OARG6	ldr r0, [fp, -OARG6]	str r0, [fp, -OARG6]
int oarg5	OARG5	add r0, fp, -OARG5	ldr r0, [fp, -OARG5]	str r0, [fp, -OARG5]

```
FP_OFF, 4
    • equ
    . equ
                     4 + FP_0FF
                     4 + I
            PF,
    • equ
            PAD,
                     0 + PF
    . equ
                     4 + PAD
            OARG6,
    •equ
            0ARG5
                     4 + 0ARG6
    . equ
            FRMADD, OARG5 - FP_OFF
    • equ
// FRMADD = 20 - 4 = 16
```

Example: Passing Stack Args, Calling Function




Variable or Argument	distanc e from fp	Address on Stack	Read variable	Write Variable
int i	I	add r0, fp, -I	ldr r0, [fp, -I]	str r0, [fp, -I]
int (*pf)()	PF	add r0, fp, -PF	ldr r0, [fp, -PF]	str r0, [fp, -PF]
int oarg6	OARG6	add r0, fp, -OARG6	ldr r0, [fp, -OARG6]	str r0, [fp, -OARG6]
int oarg5	OARG5	add r0, fp, -OARG5	ldr r0, [fp, -OARG5]	str r0, [fp, -OARG5]

```
. equ
            FP_OFF, 4
                    4 + FP 0FF
    •equ
            PF,
                    4 + I
    . equ
            PAD,
                    0 + PF
    •equ
            OARG6, 4 + PAD
    •equ
            0ARG5, 4 + 0ARG6
    •equ
            FRMADD, OARG5 - FP_OFF
    •equ
main:
            {fp, lr}
    push
    add
            fp, sp, FP_OFF
            sp, sp,-FRMADD
    add
            r0. = sum
                            // get func address
    ldr
            r1, fp, -PF
                            // PF address on stack
    add
            r0, [r1]
                            // store sum in var pf
    str
            r0, fp, −I
                            // get address of I
    add
            r1, fp, -OARG6 // address of OARG6
    add
            r0, [r1]
    str
                            // store address of I in OARG6
    ldr
            r0, [fp, -PF]
                            // get PF from stack
            r1, fp, -OARG5 // address of OARG5
    add
    str
            r0, [r1]
                            // store sum address in OARG5
            r0, 1
                            // arg 1: 1
    mov
            r1. 2
                            // arg 2: 2
    mov
            r2, 3
                            // arg 3: 3
    mov
    mov
            r3, 4
                            // arg 4: 4
    bl
            testp
    ldr
            r0, = Lmess
                            // arg 1: "%d\n"
    ldr
            r1, [fp, -I]
                            // arg 2: i
    bl
            printf
            sp, fp, FP_OFF
    sub
            {fp, lr}
    pop
    bx
```

Example: Passing Stack Args, Called Function

```
void
testp(int j, int k, int l, int m, int (*func)(int, int), int *i)
{
    *i = func(j, k) + func(l, m);
    return;
}
```



	equ equ equ	testp testp, %function FP_OFF, 20 ARG6, 8 ARG5, 4	n	
tes [.]		{r4-r7, fp, lr} fp, sp, FP_0FF		
	mov mov ldr ldr blx	r4, r2 r5, r3 r6, [fp, ARG5] r7, [fp, ARG6] r6	// load func	
	mov mov blx add str	r1, r5 r5, r0 r0, r4 r6 r0, r0, r5 r0, [r7]	<pre>// arg 2 saved m // save func() return value // arg 1 saved l // r0=func(l,m) // func(l, m) + func(j, k) // store sum to *i</pre>	
	sub pop bx	<pre>sp, fp, FP_0FF {r4-r7, fp, lr} lr</pre>		

Argument	distance from fp	Address on Stack	Read variable	Write Variable
int *i	ARG6	add r0, fp, ARG6	ldr r0, [fp, ARG6]	str r0, [fp, ARG6]
int (*fp)()	ARG5	add r0, fp, ARG5	ldr r0, [fp, ARG5]	str r0, [fp, ARG5]

Extra Slides

Determining Size of the Passed Parameter Area on The Stack

- Find the function called by main with the largest number of parameters
- That function determines the size of the Passed Parameter allocation on the stack

```
int main(void)
{
    /* code not shown */
    a(g, h);

/* code not shown */
    sixsum(a1, a2, a3, a4, a5, a6);

/* code not shown */

b(q, w, e, r);
    /* code not shown */
}
```

largest arg count is 6 allocate space for 6 - 4 = 2 arg slots

By following the saved fp, you can find each stack frame

